
1 Overview

This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2019-06_D20wk25 related to the title of this CR.

Contact:
Kalahasti, Narendra

Narendra.Kalahasti@anritsu.com
2 Corrections required
2.1 Change 1

	Function name
	fl_TC_9_1_6_1_3_TestBody

	Reason for change
	Similar change has been agreed in TTCN CR R5s200819. Deregistration procedure in step 2 is interrupted by HO and continues in step 7 with REGISTRATION REQUEST. However the RegistrationRequest after the HO may be transmitted immediately after the ReconfigurationComplete (step 6) which may collide with the HO clean-up procedures on the SS side, namely with the steps 15-16 from 38.523-3 cl. 7.3.5.3.1. Therefore it is proposed to postpone execution of the mentioned steps 15-16 until the Regestration request is received. 


	Summary of change
	Handover function is executed firstly only for steps 1-14 (38.523-3 cl. 7.3.5.3.1) and the remaining steps are executed after TC step 7.

	TTCN module
	Deregistration_NR5GC.ttcn

	MCC160 Comment
	


Before change:

  function fl_TC_9_1_6_1_3_TestBody() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);

    var NR_CellId_Type v_NGC_NASCellB := f_NR5GC_MapNASCell(ngc_CellB, NG_AllCellsOnSamePLMN);

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;//@sic R5s200350 sic@

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;//@sic R5s200350 sic@

    var template (value) RLC_Config v_Rlc_Config := cs_38508_RLC_Config_AM_DRB(cs_38508_UL_RLC_SRB, cs_38508_DL_RLC_SRB);//@sic R5s200350 sic@

    timer t_Delay := f_NR_SetTimerToleranceMin(v_NGC_NASCellA, nasTimer, 5.0);

    //Cause UE to de-register for non Switch off reason using MMI or AT commands.

    //@siclog "Step 1" siclog@

    f_UT_Deregister(UT);

    // The UE transmits DEREGISTRATION REQUEST message. (The UE starts timer T3521)

    //@siclog "Step 2" siclog@

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if(not f_Check_NG_DeregistrationReqMOMsg(v_ReceivedMsg.Pdu.Msg, crs_DeregisterType('0'B, '0'B, '01'B),f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellA, omit))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "De-registration Request Message Failed");

    }

    //The SS does not respond to DEREGISTRATION REQUEST message.

    //@siclog "Step 3" siclog@

    //The SS configures NGC Cell A as the "suitable cell" and NGC Cell B as the "Serving cell".

    //Note: T3521 value is specified as 15s in TS 24.501 [22] and it is assumed that SS canconfigure cells within this time.

    //@siclog "Step 4" siclog@

    f_NR_SetCellPower (v_NGC_NASCellA, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2);

    f_NR_SetCellPower (v_NGC_NASCellB, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    //@siclog "Step 5" siclog@

    //The SS transmits an RRCReconfiguration message on NGC Cell A to order the UE to perform intra-frequency handover to NGC Cell B.

    //@siclog "Step 6" siclog@

    //The UE transmits a RRCReconfigurationComplete message on NGC Cell B to confirm the successful completion of the intra frequency handover.

    //@sic R5s190795 sic@

    f_NR5GC_508RRC_IntraNR_HO_InterCell(v_NGC_NASCellA, v_NGC_NASCellB,

                                          -, -, -, -, -,

                                          cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount));

    //@siclog "Step 7" siclog@

    //Check: Does the UE transmit a REGISTRATION REQUEST message with registration type value set to "mobility registration updating"?

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if (not match (v_ReceivedMsg.Pdu.Msg.registration_Request.registrationType.registrationType, tsc_NG_RegistrationMobility)){// @sic R5-200208 sic@

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 7 PASS");

    }

    //@siclog "Step 8-9" siclog@

    //Steps 4-5 of the generic procedure in TS 38.508-1 [4] subclause 4.9.5.2.2  are performed.

    //@sic R5-200208 sic@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI(v_NGC_NASCellB),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellB)),// @sic R5s200350 sic@

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellB))}));// @sic R5s200350 sic@

    // Registration Accept

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellB,

                                   tsc_NR_RbId_SRB2,

                                   -,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //Registration Complete

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

    }

    //@siclog "Steps 10-22" siclog@

    //Void//@sic R5-200208 sic@

    //@siclog "Step 23" siclog@

    //Check: Does the UE transmit a DEREGISTRATION REQUEST message?

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if(not f_Check_NG_DeregistrationReqMOMsg(v_ReceivedMsg.Pdu.Msg, crs_DeregisterType('0'B, '0'B, '01'B),f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellB, omit))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "De-registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 23 PASS");

    }

    //@siclog "Step 24" siclog@

    //The SS transmits DEREGISTRATION ACCEPT message.

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellB,-, -,cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));

    //@siclog "Step 25" siclog@

    //The SS transmits an RRCRelease message.

    f_NR_RRCRelease(v_NGC_NASCellB);

    f_NR5GC_MobileInfo_Init();// @sic R5s200350 sic@

    //@siclog "Step 26" siclog@

    //Check: Does the test result of the generic procedure in TS 38.508-1 [4] subclause 4.9.13 indicates that the UE doesn't respond to paging when paged with NG-5G-S-TMSI?

    f_NR5GC_PageNoResponse(v_NGC_NASCellB);

    //@siclog "Step 27" siclog@

    //Cause UE to initiate registration using MMI or AT commands.

    f_UT_Register(UT);

    //@siclog "Step 28" siclog@

    //The UE performs Registration procedure as specified in TS 38.508-1 [4] subclause 4.5.2 with 'connected without release'.

    f_NR5GC_RRC_IdleState1N_Def(v_NGC_NASCellB, -, noRrcConnectionRelease);

    //@siclog "Step 28A - 28B" siclog@//@sic R5s200350 sic@

    //SS transmits NR RRCReconfigurationmessage to configure specific maxRetxThreshold for NGC Cell B.

    //The UE transmits NR RRCReconfigurationComplete message.

    v_Rlc_Config.am.ul_AM_RLC.maxRetxThreshold := t32;

    v_RLC_BearerConfigList[0] := valueof(cs_NR_BearerConfigDef(f_NR_LogicalChannelId_SRB(tsc_SRB2),

                                                               cs_NR_ServedRadioBearerSRB(tsc_SRB2),

                                                               v_Rlc_Config,

                                                               cs_38508_LogicalChannelConfig_SRB(tsc_SRB2)));

    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(cs_38508_CellGroupConfig(tsc_NR_CellGroupId_MCG, v_RLC_BearerConfigList))));

    f_NR_SendRRCReconfiguration (v_NGC_NASCellB, -,cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB2)}), v_V1530Ext,-,-,-,true);

    //@siclog "Step 29" siclog@

    //The SS is configured on NGC Cell B not to send RLC acknowledgements (RLC ACKs) to the UE (Note1).

    f_NR_SS_ConfigRB_AckProhibit(v_NGC_NASCellB,tsc_NR_RbId_SRB2,Prohibit);//@sic R5-197087 sic@

    //@siclog "Step 30" siclog@

    //If possible (see ICS) switch off is performed or the USIM is removed. Otherwise the power is removed.

    f_UT_SwitchOffUE(UT, false);

    if(pc_SwitchOnOff or pc_USIM_Removal){// @sic R5-202660 sic@

    //@siclog "Step 31" siclog@

    //The UE transmits DEREGISTRATION REQUEST message.

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if(not f_Check_NG_DeregistrationReqMOMsg(v_ReceivedMsg.Pdu.Msg, crs_DeregisterType('1'B, '0'B, '01'B),f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellB, omit))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "De-registration Request Message Failed");

    }

    //@siclog "Step 32" siclog@

    //Void @sic R5-198185 sic@

    //@siclog "Step 33" siclog@

    //The SS transmits anRRCReconfiguration message on NGC Cell B to order the UE to perform intra-frequency handover to NGC Cell A.

    //@siclog "Step 34" siclog@

    //UE transmit a RRCReconfigurationComplete message on NGC Cell A to confirm the successful completion of the intra frequency handover (Note 2).

    f_NR_ULGrantConfiguration_Start(v_NGC_NASCellA);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(v_NGC_NASCellB, v_NGC_NASCellA,//@sic R5-197222 sic@

                                          -, -, -, -, -,

                                          cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount));

    //@siclog "Step 35" siclog@

    //Check: Does the UE transmit a REGISTRATION REQUEST message with registration type value set to "mobility registration updating"? within 5 seconds?

    t_Delay.start;

    alt {

      [] SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA,tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered)))-> value v_ReceivedAsp//@sic R5s200561 sic@

        {

          v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

          if (not match (v_ReceivedMsg.Pdu.Msg.registration_Request.registrationType.registrationType, tsc_NG_RegistrationMobility)){// @sic R5-200208 sic@

            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 35: REGISTRATION REQUEST received");

            t_Delay.stop;

          }

        }

      [] t_Delay.timeout{}

    }

   }

  }
After change:

  function fl_TC_9_1_6_1_3_TestBody() runs on NR5GC_PTC

  {

    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);

    var NR_CellId_Type v_NGC_NASCellB := f_NR5GC_MapNASCell(ngc_CellB, NG_AllCellsOnSamePLMN);

    var NR_SRB_COMMON_IND v_ReceivedAsp;

    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;

    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;

    var template (omit) RRCReconfiguration_v1530_IEs v_V1530Ext;//@sic R5s200350 sic@

    var template (value) NR_RLC_BearerToAddModList_Type v_RLC_BearerConfigList;//@sic R5s200350 sic@

    var template (value) RLC_Config v_Rlc_Config := cs_38508_RLC_Config_AM_DRB(cs_38508_UL_RLC_SRB, cs_38508_DL_RLC_SRB);//@sic R5s200350 sic@

    timer t_Delay := f_NR_SetTimerToleranceMin(v_NGC_NASCellA, nasTimer, 5.0);

    //Cause UE to de-register for non Switch off reason using MMI or AT commands.

    //@siclog "Step 1" siclog@

    f_UT_Deregister(UT);

    // The UE transmits DEREGISTRATION REQUEST message. (The UE starts timer T3521)

    //@siclog "Step 2" siclog@

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if(not f_Check_NG_DeregistrationReqMOMsg(v_ReceivedMsg.Pdu.Msg, crs_DeregisterType('0'B, '0'B, '01'B),f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellA, omit))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "De-registration Request Message Failed");

    }

    //The SS does not respond to DEREGISTRATION REQUEST message.

    //@siclog "Step 3" siclog@

    //The SS configures NGC Cell A as the "suitable cell" and NGC Cell B as the "Serving cell".

    //Note: T3521 value is specified as 15s in TS 24.501 [22] and it is assumed that SS canconfigure cells within this time.

    //@siclog "Step 4" siclog@

    f_NR_SetCellPower (v_NGC_NASCellA, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR1, tsc_NR_SuitableIntraFreqCellSSS_EPRE_FR2);

    f_NR_SetCellPower (v_NGC_NASCellB, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2);

    //@siclog "Step 5" siclog@

    //The SS transmits an RRCReconfiguration message on NGC Cell A to order the UE to perform intra-frequency handover to NGC Cell B.

    //@siclog "Step 6" siclog@

    //The UE transmits a RRCReconfigurationComplete message on NGC Cell B to confirm the successful completion of the intra frequency handover.

    //@sic R5s190795 sic@   

    f_NR5GC_508RRC_IntraNR_HO_InterCell_Steps1_14(v_NGC_NASCellA, v_NGC_NASCellB,

                                          -, -, -, -, -,

                                          cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount));

    //@siclog "Step 7" siclog@

    //Check: Does the UE transmit a REGISTRATION REQUEST message with registration type value set to "mobility registration updating"?

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if (not match (v_ReceivedMsg.Pdu.Msg.registration_Request.registrationType.registrationType, tsc_NG_RegistrationMobility)){// @sic R5-200208 sic@

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 7 PASS");

    }

  //The SS configuration after HO can now be finished

  f_NR5GC_508RRC_IntraNR_HO_InterCell_Common_Steps15_16(v_NGC_NASCellA);
    //@siclog "Step 8-9" siclog@

    //Steps 4-5 of the generic procedure in TS 38.508-1 [4] subclause 4.9.5.2.2  are performed.

    //@sic R5-200208 sic@

    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Mobility,

                                                   f_NR5GC_GetMobileIdGUTI(v_NGC_NASCellB),

                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (v_NGC_NASCellB)),// @sic R5s200350 sic@

                                                                                {bit2oct(f_NR_CellInfo_GetTAC(v_NGC_NASCellB))}));// @sic R5s200350 sic@

    // Registration Accept

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellB,

                                   tsc_NR_RbId_SRB2,

                                   -,

                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));

    //Registration Complete

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    if (not match (v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg, cr_NG_REGISTRATION_COMPLETE)) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Registration Complete Message Failed");

    }

    //@siclog "Steps 10-22" siclog@

    //Void//@sic R5-200208 sic@

    //@siclog "Step 23" siclog@

    //Check: Does the UE transmit a DEREGISTRATION REQUEST message?

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if(not f_Check_NG_DeregistrationReqMOMsg(v_ReceivedMsg.Pdu.Msg, crs_DeregisterType('0'B, '0'B, '01'B),f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellB, omit))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "De-registration Request Message Failed");

    }

    else {

      f_NR_PreliminaryPass (__FILE__, __LINE__, "Step 23 PASS");

    }

    //@siclog "Step 24" siclog@

    //The SS transmits DEREGISTRATION ACCEPT message.

    SRB.send(cas_NR_SRB_NasPdu_REQ(v_NGC_NASCellB,-, -,cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_NG_DEREGISTRATION_ACCEPT)));

    //@siclog "Step 25" siclog@

    //The SS transmits an RRCRelease message.

    f_NR_RRCRelease(v_NGC_NASCellB);

    f_NR5GC_MobileInfo_Init();// @sic R5s200350 sic@

    //@siclog "Step 26" siclog@

    //Check: Does the test result of the generic procedure in TS 38.508-1 [4] subclause 4.9.13 indicates that the UE doesn't respond to paging when paged with NG-5G-S-TMSI?

    f_NR5GC_PageNoResponse(v_NGC_NASCellB);

    //@siclog "Step 27" siclog@

    //Cause UE to initiate registration using MMI or AT commands.

    f_UT_Register(UT);

    //@siclog "Step 28" siclog@

    //The UE performs Registration procedure as specified in TS 38.508-1 [4] subclause 4.5.2 with 'connected without release'.

    f_NR5GC_RRC_IdleState1N_Def(v_NGC_NASCellB, -, noRrcConnectionRelease);

    //@siclog "Step 28A - 28B" siclog@//@sic R5s200350 sic@

    //SS transmits NR RRCReconfigurationmessage to configure specific maxRetxThreshold for NGC Cell B.

    //The UE transmits NR RRCReconfigurationComplete message.

    v_Rlc_Config.am.ul_AM_RLC.maxRetxThreshold := t32;

    v_RLC_BearerConfigList[0] := valueof(cs_NR_BearerConfigDef(f_NR_LogicalChannelId_SRB(tsc_SRB2),

                                                               cs_NR_ServedRadioBearerSRB(tsc_SRB2),

                                                               v_Rlc_Config,

                                                               cs_38508_LogicalChannelConfig_SRB(tsc_SRB2)));

    v_V1530Ext := cs_NR_RRCReconfiguration_v1530_IEs(bit2oct(encvalue(cs_38508_CellGroupConfig(tsc_NR_CellGroupId_MCG, v_RLC_BearerConfigList))));

    f_NR_SendRRCReconfiguration (v_NGC_NASCellB, -,cs_38508_RadioBearerConfigDef({cs_NR_SRB_ToAddMod(tsc_SRB2)}), v_V1530Ext,-,-,-,true);

    //@siclog "Step 29" siclog@

    //The SS is configured on NGC Cell B not to send RLC acknowledgements (RLC ACKs) to the UE (Note1).

    f_NR_SS_ConfigRB_AckProhibit(v_NGC_NASCellB,tsc_NR_RbId_SRB2,Prohibit);//@sic R5-197087 sic@

    //@siclog "Step 30" siclog@

    //If possible (see ICS) switch off is performed or the USIM is removed. Otherwise the power is removed.

    f_UT_SwitchOffUE(UT, false);

    if(pc_SwitchOnOff or pc_USIM_Removal){// @sic R5-202660 sic@

    //@siclog "Step 31" siclog@

    //The UE transmits DEREGISTRATION REQUEST message.

    SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellB, tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp;

    v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

    if(not f_Check_NG_DeregistrationReqMOMsg(v_ReceivedMsg.Pdu.Msg, crs_DeregisterType('1'B, '0'B, '01'B),f_NR5GC_GetMobileIdGUTI (v_NGC_NASCellB, omit))) {

      f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "De-registration Request Message Failed");

    }

    //@siclog "Step 32" siclog@

    //Void @sic R5-198185 sic@

    //@siclog "Step 33" siclog@

    //The SS transmits anRRCReconfiguration message on NGC Cell B to order the UE to perform intra-frequency handover to NGC Cell A.

    //@siclog "Step 34" siclog@

    //UE transmit a RRCReconfigurationComplete message on NGC Cell A to confirm the successful completion of the intra frequency handover (Note 2).

    f_NR_ULGrantConfiguration_Start(v_NGC_NASCellA);

    f_NR5GC_508RRC_IntraNR_HO_InterCell(v_NGC_NASCellB, v_NGC_NASCellA,//@sic R5-197222 sic@

                                          -, -, -, -, -,

                                          cs_MasterKeyUpdate(false, tsc_NR_38508_NextHopChainingCount));

    //@siclog "Step 35" siclog@

    //Check: Does the UE transmit a REGISTRATION REQUEST message with registration type value set to "mobility registration updating"? within 5 seconds?

    t_Delay.start;

    alt {

      [] SRB.receive(car_NR_SRB_NasPdu_IND(v_NGC_NASCellA,tsc_NR_RbId_SRB2, cr_NG_NAS_Ind(tsc_SHT_IntegrityProtected_Ciphered)))-> value v_ReceivedAsp//@sic R5s200561 sic@

        {

          v_ReceivedMsg := v_ReceivedAsp.Signalling.Nas[0];

          if (not match (v_ReceivedMsg.Pdu.Msg.registration_Request.registrationType.registrationType, tsc_NG_RegistrationMobility)){// @sic R5-200208 sic@

            f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 35: REGISTRATION REQUEST received");

            t_Delay.stop;

          }

        }

      [] t_Delay.timeout{}

    }

   }

  }
